二変数の極限

非孤立特異点を持つ二変数の有理関数の極限について、limit コマンドが改善されました。従来のリリースではこのような極限を特定することができませんでしたが、今回の改善により計算できるようになりました。このような状況で極限が存在する場合、この極限は $+\infty$ または $-\infty$ と表されます。また、Maple では、この極限が存在しないことを特定することもでき、この場合は *undefined* を返します。

Maple 18 では、以下のすべての limit コマンドで結果を得ることができませんでしたが、Maple 2015 では可能になりました。

> $f := \frac{x\,y}{x+y}$:

> $\mathit{limit}(f, \{x=0, y=0\})$

$$\mathrm{undefined} \tag{1}$$

> $g := \frac{4\,x\,y}{(x-y)^2}$:

> $\mathit{limit}(g, \{x=0, y=0\})$

$$\mathrm{undefined} \tag{2}$$

> $h := \frac{x^4 - x^2 - y^2}{(x-y)^4}$:

> $\mathit{limit}(h, \{x=0, y=0\})$

$$-\infty \tag{3}$$

これら 3 つの関数を原点近傍でプロットしてみましょう。

最初の例では、以下のプロットに示すように、f は特異点 $y=-x$ の片方では $+\infty$、反対側では $-\infty$ に収束します。そのため、原点に極限は存在しません。

> $\mathit{pf1} := \mathit{plot3d}(\mathit{piecewise}(y+x<0, f, \mathit{undefined}), x=-0.1..0.1, y=-0.1..0.1, \mathit{axes}=\mathit{boxed}, \mathit{view}=-10..10, \mathit{numpoints}=50000)$:

> $\mathit{pf2} := \mathit{plot3d}(\mathit{piecewise}(y+x>0, f, \mathit{undefined}), x=-0.1..0.1, y=-0.1..0.1, \mathit{axes}=\mathit{boxed}, \mathit{view}=-10..10, \mathit{numpoints}=40000)$:

> $\mathit{plots}\text{:-}\mathit{display}(\mathit{pf1}, \mathit{pf2})$



次に、2 つ目の例について考えてみましょう。

```
> s := plots:-spacecurve([x,-x,-1], x=-0.1..0.1, color=red, thickness=3):
  pg := plot3d(g, x=-0.1..0.1, y=-0.1..0.1, axes=boxed, view=-10..100, numpoints=10000):
  plots:-display(s, pg)
```



g は、特異点 $y=x$ の近傍で $+\infty$ に収束します。ただし、反対角 $y=-x$ に沿って、極限は有限となります。

```
> eval(g, y=-x)
```

$$-1 \tag{4}$$

このため、g は原点で極限を持ちません。実際に、≥ -1 の任意の数字が、$-1 \leq a < 1$ の半直線 $y=ax$ に沿って極限として現れます。

```
> eval(g, y=a x)
```

$$\frac{4x^2 a}{(-ax+x)^2} \tag{5}$$

```
> limit((5), x=0)
```

$$\frac{4a}{(a-1)^2} \tag{6}$$

```
> plot((6), a=-3..1.1, view=-2..10)
```



最後の例では、特異点 $y = x$ の両側で、h は $-\infty$ に収束します。

> $plot3d(h, x = -0.1\,..0.1, y = -0.1\,..0.1, axes = boxed, view = -100000\,..1)$



ただし、この場合、$a \neq 1$ の任意の半直線 $y = a\,x$ に沿った極限は、$-\infty$ でもあります。

> $eval(h, y = a\,x)$

$$\frac{-a^2 x^2 + x^4 - x^2}{(-a\,x + x)^4} \tag{7}$$

> $limit(\mathbf{(7)}, x = 0)$

$$-\mathrm{signum}\left(\frac{a^2 + 1}{a^4 - 4\,a^3 + 6\,a^2 - 4\,a + 1}\right)\,\infty \tag{8}$$

> $factor(\mathbf{(8)})$

$$-\mathrm{signum}\left(\frac{a^2 + 1}{(a - 1)^4}\right)\,\infty \tag{9}$$

> $simplify(\mathbf{(9)})$ assuming $a < 1$;
> $simplify(\mathbf{(9)})$ assuming $a > 1$

$$\begin{gathered} -\infty \\ -\infty \end{gathered} \tag{10}$$

極限が存在すること、そしてその極限が原点に近づく任意の曲線で $-\infty$ であることを、ラグランジュ未定乗数法を使用して証明できます。固定半径 r の円上の関数 h の極値 (最大値および最小値) では、関数の勾配と円の制約式の勾配が平行であるという条件を満たします。

> $C := x^2 + y^2 - r^2 :$

> $with(VectorCalculus) :$

> $df := normal\sim(Jacobian([h], [x, y]))$

$$df := \left[\begin{array}{cc} -\frac{2\,(2\,x^3\,y - x^2 - x\,y - 2\,y^2)}{(x - y)^5} & \frac{2\,(2\,x^4 - 2\,x^2 - x\,y - y^2)}{(x - y)^5} \end{array}\right] \tag{11}$$

> $dC := Jacobian([C], [x, y])$

$$dC := \left[\begin{array}{cc} 2\,x & 2\,y \end{array}\right] \tag{12}$$

> $eq := factor\left(numer\left(normal\left(df_{1,1}\,dC_{1,2} - df_{1,2}\,dC_{1,1}\right)\right)\right)$

$$eq := -8\,\left(x^3 - x - y\right)\,\left(x^2 + y^2\right) \tag{13}$$

したがって、C における f の局所的な最大値と最小値は $C=0$ と $eq=0$ の両方が成り立つ場合に発生します。つまり、二変数の極限では、$eq=0$ を満たすクリティカルパスのみを考慮すれば十分であることを意味します。ただし、特異点 $y=x$ の近傍で発生する可能性がある全域的な上限および下限についても考慮する必要があります。
この例では、eq の因数 x^2+y^2 ではいかなる実際のパスも受け入れないため、$x^3-x-y=0$ (または、等価な $y=-x+x^3$) で求められるクリティカルパスが 1 つだけ存在します。

> $normal(eval(h, y=-x+x^3))$

$$-\frac{x^2-1}{x^2\left(x^2-2\right)^3} \tag{14}$$

> $limit(\mathbf{(14)}, x=0)$

$$-\infty \tag{15}$$

さらに特異点 $y=x$ に近い極限を証明するためには、特異点に沿った極限をとることはできません。代わりに、以下にそれぞれ示すように、上側および下側から近接して特異点に近づく 2 つの曲線について考えます。

> $c1 := y=x+x^2$

$$c1 := y=x^2+x \tag{16}$$

> $c2 := y=x-x^2$

$$c2 := y=-x^2+x \tag{17}$$

> $plot([x, rhs(c1), rhs(c2), -x+x^3], x=-0.5..0.5,$
$legend=$ ["singular path", "c1", "c2", "critical path"])



> $normal(eval(h, c1))$

$$-\frac{2\,(x+1)}{x^6} \tag{18}$$

> $limit(\mathbf{(18)}, x=0)$

$$-\infty \tag{19}$$

> $normal(eval(h, c2))$

$$\frac{2\,(x-1)}{x^6} \tag{20}$$

> $limit(\mathbf{(20)}, x=0)$

$$-\infty \tag{21}$$

参照

updates/Maple17/BivariateLimits、極値、limit/multi - 多次元の極限